

Support: Get Support Information

Support

Functions

afx_msg VARIANT	OSSupportUI::GetSupportCount () Gets the total number of supported nodes exist in the current structure.
afx_msg VARIANT	OSSupportUI::GetSupportNodes (VARIANT FAR &varNodeNos) Gets all supported nodes in an array.
afx_msg VARIANT	OSSupportUI::GetSupportType (VARIANT FAR &vaSupportNode) Gets the support type for the specified node.
afx_msg VARIANT	OSSupportUI::GetSupportInformation (const VARIANT FAR &vaSupportNode, VARIANT FAR &varReleaseSpec, VARIANT FAR &varSpringSpec) Gets support information for the specified node, including release specification and spring specification.
afx_msg VARIANT	OSSupportUI::GetSupportInformationEx (const VARIANT FAR &vaSupportNode, VARIANT FAR &vaSupportNo, VARIANT FAR &vaSupportType, VARIANT FAR &varReleaseSpec, VARIANT FAR &varSpringSpec) Gets support information for the specified node, including release specification and spring specification.
afx_msg VARIANT	OSSupportUI::GetCountOfElasticMat () Gets the total number of ElasticMat supports in the current structure.
afx_msg VARIANT	OSSupportUI::GetElasticMatDetail (const VARIANT FAR &supportid, VARIANT FAR &dir, VARIANT FAR &subgrade, VARIANT FAR &print, VARIANT FAR &springType, VARIANT FAR &nodesCount) Get elastic mat support information for a specific support Id.
afx_msg VARIANT	OSSupportUI::GetElasticMatAssignmentList (const VARIANT FAR &supportid, VARIANT FAR &nodeList) Get elastic mat support entity list for a specific support Id.
afx_msg VARIANT	OSSupportUI::GetCountOfPlateMat () Gets the total number of plate mat supports in the current structure.
afx_msg VARIANT	OSSupportUI::GetPlateMatSupportId (const VARIANT FAR &varnPlateMatIndex) Gets the plate mat support ID.
afx_msg VARIANT	OSSupportUI::GetPlateMatDetail (const VARIANT FAR &varnPlateMatNo, VARIANT FAR &varDirection, VARIANT FAR &varSubgrade1, VARIANT FAR &varSubgrade2, VARIANT FAR &varSubgrade3, VARIANT FAR &varPrint, VARIANT FAR &varSpringType, VARIANT FAR &nAssignedPlateCount) Get plate mat support information for a specific support Id.
afx_msg VARIANT	OSSupportUI::GetPlateMatAssignmentList (const VARIANT FAR &varnPlateMatNo, VARIANT FAR &varPlateNos) Get elastic mat support entity list for a specific support Id.
afx_msg VARIANT	OSSupportUI::GetCountOfElasticFooting () Gets the total number of elastic footing supports in the current structure.
afx_msg VARIANT	OSSupportUI::GetElasticFootingDetail (const VARIANT FAR &supportid, VARIANT FAR &length, VARIANT FAR &width, VARIANT FAR &dir, VARIANT FAR &subgrade, VARIANT FAR &nodesCount) Get elastic footing support information for a specific support Id.
afx_msg VARIANT	OSSupportUI::GetElasticFootingAssignmentList (const VARIANT FAR &supportid, VARIANT FAR &nodeList) Get list of assigned node Id for a specific elastic footing support Id.

Detailed Description

These functions are related to get support Information. @{@}

Function Documentation

◆ GetCountOfElasticFooting()

VARIANT OSSupportUI::GetCountOfElasticFooting ()

Gets the total number of elastic footing supports in the current structure.

Returns

The number of elastic footing supports.

Example: VBA

```
Option Explicit

Sub Main
    Dim objOpenStaad As Object
    Dim stdFile As String
    Dim Result As Long

    Set objOpenStaad = GetObject(,"StaadPro.OpenSTAAD")
    objOpenStaad.GetSTAADFile stdFile, "TRUE"
    If stdFile="" Then
        MsgBox"Bad"
        Set objOpenStaad = Nothing
        Exit Sub
    End If

    Result = objOpenStaad.Support.GetCountOfElasticFooting()

    If Result > 0 Then
        MsgBox"Elastic footing found."
    Else
        MsgBox"Elastic footing not found."
    End If

    Set objOpenStaad = Nothing
End Sub
```

◆ GetCountOfElasticMat()

VARIANT OSSupportUI::GetCountOfElasticMat ()

Gets the total number of ElasticMat supports in the current structure.

Returns

The number of ElasticMat supports.

Example: VBA

```
Option Explicit

Sub Main
    Dim objOpenStaad As Object
    Dim stdFile As String
    Dim Result As Long

    Set objOpenStaad = GetObject(,"StaadPro.OpenSTAAD")
    objOpenStaad.GetSTAADFile stdFile, "TRUE"
    Result = objOpenStaad.Support.GetCountOfElasticMat()
End Sub
```

◆ GetCountOfPlateMat()

VARIANT OSSupportUI::GetCountOfPlateMat ()

Gets the total number of plate mat supports in the current structure.

Returns

The number of plate mat supports.

Example: VBA

```
Option Explicit
Sub Main
    Dim objOpenStaad As Object
    Dim stdFile As String
    Set objOpenStaad = GetObject(,"StaadPro.OpenSTAAD")
    objOpenStaad.GetSTAADFile stdFile, "TRUE"
    If stdFile="" Then
        MsgBox"Bad"
        Set objOpenStaad = Nothing
        Exit Sub
    End If
    Dim resVal As Long
    resVal = objOpenStaad.Support.GetCountOfPlateMat()
    MsgBox"Plate Mat Count" +Str (resVal)
    Set objOpenStaad = Nothing
End Sub
```

◆ GetElasticFootingAssignmentList()

VARIANT OSSupportUI::GetElasticFootingAssignmentList (const VARIANT FAR & supportid,
 VARIANT FAR & nodeList)

Get list of assigned node Id for a specific elastic footing support Id.

Parameters

[in] **supportid** Support reference ID (Type: Long).

[out] **nodeList** List of assigned node ID (Type: Long).

Return values

1 if the method is successful.

0 if the method is unsuccessful.

Example: VBA

```
Option Explicit

Sub Main
    Dim objOpenStaad As Object
    Dim stdFile As String

    Set objOpenStaad = GetObject(, "StaadPro.OpenSTAAD")
    objOpenStaad.GetSTAADFile stdFile, "TRUE"
    If stdFile="" Then
        MsgBox "Bad"
        Set objOpenStaad = Nothing
        Exit Sub
    End If

    Dim FootingCount As Long
    FootingCount = objOpenStaad.Support.GetCountOfElasticFooting()
    If FootingCount > 0 Then
        Dim Result As Long
        Dim Length As Double
        Dim Width As Double
        Dim Direction As Long
        Dim Subgrade As Double
        Dim NodeCount As Long
        Dim NodeList() As Long
        Result = objOpenStaad.Support.GetElasticFootingDetail(3, Length, Width, Direction, Subgrade,
        NodeCount)
        If Result = 1 Then
            If NodeCount > 0 Then
                ReDim NodeList(NodeCount - 1) As Long
                Dim Success As Long
                Success = objOpenStaad.Support.GetElasticFootingAssignmentList(3, NodeList)
                If Success = 1 Then
                    MsgBox "Elastic footing assigned node list found."
                Else
                    MsgBox "Elastic footing assigned node list not found."
                End If
            Else
                MsgBox "Elastic footing not assigned."
            End If
        Else
            MsgBox "Elastic footing details not found."
        End If
    Else
        MsgBox "Elastic footing not found."
    End If

    Set objOpenStaad = Nothing
End Sub
```

◆ GetElasticFootingDetail()

VARIANT OSSupportUI::GetElasticFootingDetail (const VARIANT FAR & **supportid**,
 VARIANT FAR & **length**,
 VARIANT FAR & **width**,
 VARIANT FAR & **dir**,
 VARIANT FAR & **subgrade**,
 VARIANT FAR & **nodesCount**)

Get elastic footing support information for a specific support Id.

Parameters

- [in] **supportid** Support reference ID (Type: Long).
 [out] **length** Length of footing (Type: Double).
 [out] **width** Width of footing (Type: Double).
 [out] **dir** Direction of the resistance of the spring supports (Type: Long):

Value	Direction
0	X Direction
1	Y Direction
2	Z Direction
3	X Only Direction
4	Y Only Direction
5	Z Only Direction

- [out] **subgrade** Subgrade Modulus of the soil (Type: Double).
 [out] **nodesCount** Count of assigned nodes (Type: Long).

Return values

- 1** if the method is successful.
0 if the method is unsuccessful.

Example: VBA

```
Option Explicit

Sub Main
    Dim objOpenStaad As Object
    Dim stdFile As String

    Set objOpenStaad = GetObject("StaadPro.OpenSTAAD")
    objOpenStaad.GetSTAADFile stdFile, "TRUE"
    If stdFile="" Then
        MsgBox"Bad"
        Set objOpenStaad = Nothing
        Exit Sub
    End If

    Dim FootingCount As Long
    FootingCount = objOpenStaad.Support.GetCountOfElasticFooting()
    if FootingCount > 0 Then
        Dim Result As Long
        Dim Length As Double
        Dim Width As Double
        Dim Direction As Long
        Dim Subgrade As Double
        Dim NodeCount As Long
        Result = objOpenStaad.Support.GetElasticFootingDetail(2, Length, Width, Direction, Subgrade,
        NodeCount)
        If Result = 1 Then
            MsgBox"Elastic footing details found."
        Else
```

```

        MsgBox"Elastic footing details not found."
    End If
Else
    MsgBox"Elastic footing not found."
End If

Set objOpenStaad = Nothing
End Sub

```

◆ GetElasticMatAssignmentList()

VARIANT OSSupportUI::GetElasticMatAssignmentList (const VARIANT FAR & supportid,
 VARIANT FAR & nodeList)

Get elastic mat support entity list for a specific support Id.

Parameters

[in] **supportid** (Type: Long).
 [out] **nodeList** (Type: Long).

Return values

TRUE Successful.
FALSE Unsuccessful.

Example: VBA

Option Explicit

```

Sub Main
    Dim objOpenStaad As Object
    Dim stdFile As String
    Dim supportId As Long
    Dim dir As Long
    Dim subgrade As Double
    Dim print As Long
    Dim springType As Long
    Dim nodes As Long
    Dim Result1 As Boolean
    Dim Result2 As Boolean
    Dim nodeList() As Long

    Set objOpenStaad = GetObject(,"StaadPro.OpenSTAAD")
    objOpenStaad.GetSTAADFile stdFile, "TRUE"
    supportId = 4
    Result1 = objOpenStaad.support.GetElasticMatDetail(supportId,dir,subgrade,print,springType,nodes)
    If nodes > 0 Then
        ReDim nodeList(nodes - 1) As Long
        Result2 = objOpenStaad.support.GetElasticMatAssignmentList(supportId,nodeList)
    End If
End Sub

```

◆ GetElasticMatDetail()

```
VARIANT OSSupportUI::GetElasticMatDetail ( const VARIANT FAR & supportid,
                                           VARIANT FAR & dir,
                                           VARIANT FAR & subgrade,
                                           VARIANT FAR & print,
                                           VARIANT FAR & springType,
                                           VARIANT FAR & nodesCount )
```

Get elastic mat support information for a specific support Id.

Parameters

[in] **supportid** (Type: Long).

[out] **dir** (Type: Long):

Value	Inclined Type
0	X Direction
1	Y Direction
2	Z Direction
3	X Only Direction
4	Y Only Direction
5	Z Only Direction

[out] **subgrade** Subgrade Modulus (Type: Double)

[out] **print** (Type: Long)

Value	Print Influence
0	Not Checked
1	Checked

[out] **springType** Generate Compression (Type: Long)

Value	Spring Type
0	None
1	Compression Only
2	Multi-Linear

[out] **nodesCount** (Type: Long).

Return values

TRUE Successful.

FALSE Unsuccessful.

Example: VBA

```
Option Explicit

Sub Main
    Dim objOpenStaad As Object
    Dim stdFile As String
    Dim supportId As Long
    Dim dir As Long
    Dim subgrade As Double
    Dim print As Long
    Dim springType As Long
    Dim nodes As Long
    Dim Result As Boolean

    Set objOpenStaad = GetObject(,"StaadPro.OpenSTAAD")
    objOpenStaad.GetSTAADFile stdFile, "TRUE"
```

```

supportId = 4
Result = objOpenStaad.support.GetElasticMatDetail(supportId,dir,subgrade,print,springType,nodes)
End Sub

```

◆ GetPlateMatAssignmentList()

VARIANT OSSupportUI::GetPlateMatAssignmentList (const VARIANT FAR & **varnPlateMatNo**,
 VARIANT FAR & **varPlateNos**)

Get elastic mat support entity list for a specific support Id.

Parameters

[in] **supportid** (Type: Long).
 [out] **nodeList** (Type: Long).

Return values

TRUE Successful.
FALSE Unsuccessful.

Example: VBA

```

Option Explicit
Sub Main
  Dim objOpenStaad As Object
  Dim stdFile As String
  Set objOpenStaad = GetObject("StaadPro.OpenSTAAD")
  objOpenStaad.GetSTAADFile stdFile, "TRUE"
  If stdFile="" Then
    MsgBox"Bad"
    Set objOpenStaad = Nothing
    Exit Sub
  End If
  Dim supportIndex As Long
  Dim supportId As Long
  supportIndex = 1
  supportId = objOpenStaad.Support.GetPlateMatSupportId(supportIndex)
  Dim dir As Long
  Dim subgrade As Double
  Dim print As Long
  Dim springType As Long
  Dim bRes As Boolean
  Dim nPlateCount As Long
  bRes= objOpenStaad.support.GetPlateMatDetail(supportId,dir,subgrade,print,springType,nPlateCount)
  If(Not bRes)Then
    MsgBox"GetPlateMatDetail Error"
    Exit Sub
  End If
  Dim plateList() As Long
  ReDim plateList(nPlateCount) As Long
  bRes= objOpenStaad.support.GetPlateMatAssignmentList(supportId, plateList)
  If(bRes)Then
    MsgBox"successful"
  Else
    MsgBox"Error"
  End If
  Set objOpenStaad = Nothing
End Sub

```

◆ GetPlateMatDetail()


```

VARIANT OSSupportUI::GetPlateMatDetail ( const VARIANT FAR & varnPlateMatNo,
                                         VARIANT FAR & varDirection,
                                         VARIANT FAR & varSubgrade1,
                                         VARIANT FAR & varSubgrade2,
                                         VARIANT FAR & varSubgrade3,
                                         VARIANT FAR & varPrint,
                                         VARIANT FAR & varSpringType,
                                         VARIANT FAR & nAssignedPlateCount )

```

Get plate mat support information for a specific support Id.

Parameters

[in] **varnPlateMatNo** (Type: Long).

[out] **varDirection** (Type: Long):

Value	Inclined Type
0	X Direction
1	Y Direction
2	Z Direction
3	X Only Direction
4	Y Only Direction
5	Z Only Direction
6	All Direction

[out] **varSubgrade** Subgrade Modulus (Type: Double)

[out] **varPrint** (Type: Long)

Value	Print Influence
0	Not Checked
1	Checked

[out] **varSpringType** Generate Compression (Type: Long)

Value	Spring Type
0	None
1	Compression Only
2	Multi-Linear

[out] **nAssignedPlateCount** (Type: Long).

Return values

TRUE Successful.

FALSE Unsuccessful.

Example: VBA

```

Option Explicit
Sub Main
    Dim objOpenStaad As Object
    Dim stdFile As String
    Set objOpenStaad = GetObject(, "StaadPro.OpenSTAAD")
    objOpenStaad.GetSTAADFile stdFile, "TRUE"
    If stdFile="" Then
        MsgBox "Bad"
        Set objOpenStaad = Nothing
    Exit Sub

```

```

End If
Dim supportIndex As Long
Dim supportId As Long
supportIndex = 1
supportId = objOpenStaad.Support.GetPlateMatSupportId(supportIndex)
Dim dir As Long
Dim subgrade1 As Double
Dim subgrade2 As Double
Dim subgrade3 As Double
Dim print As Long
Dim springType As Long
Dim bRes As Boolean
Dim nPlateCount As Long
bRes=
objOpenStaad.support.GetPlateMatDetail(supportId,dir,subgrade1,subgrade2,subgrade3,print,springType,nPlateCo
If(bRes)Then
    MsgBox"successful"
Else
    MsgBox"Error"
End If
Set objOpenStaad = Nothing
End Sub

```

◆ GetPlateMatSupportId()

VARIANT OSSupportUI::GetPlateMatSupportId (const VARIANT FAR & varnPlateMatIndex)

Gets the plate mat support ID.

Parameters

[in] **varnPlateMatIndex** PlateMat Index, start from 0(Type: Long).

Returns

The number of plate mat supports, -1 - invalid value for varnPlateMatIndex

Example: VBA

```

Option Explicit
Sub Main
    Dim objOpenStaad As Object
    Dim stdFile As String
    Set objOpenStaad = GetObject(,"StaadPro.OpenSTAAD")
    objOpenStaad.GetSTAADFile stdFile, "TRUE"
    If stdFile="" Then
        MsgBox"Bad"
        Set objOpenStaad = Nothing
        Exit Sub
    End If
    Dim resVal As Long
    resVal = objOpenStaad.Support.GetPlateMatSupportId(0)
    MsgBox"The first Plate Mat Count is " +Str (resVal)
    Set objOpenStaad = Nothing
End Sub

```

◆ GetSupportCount()

VARIANT OSSupportUI::GetSupportCount ()

Gets the total number of supported nodes exist in the current structure.

Returns

The number of support.

Example: VBA

```
Sub SupportCount()
    Dim objOpenSTAAD As Object
    Dim SelNodesNo As Long
    Dim SelNodes() As Long
    Dim pinnedSupport As Long
    Dim assignSupport As Variant
    Dim supportCount As Variant
    Dim I As Integer

    'Launch OpenSTAAD Object
    Set objOpenSTAAD = GetObject(, "StaadPro.OpenSTAAD")

    'Get no. of selected nodes
    SelNodesNo = objOpenSTAAD.Geometry.GetNoOfSelectedNodes

    'Create pinned support
    pinnedSupport = objOpenSTAAD.Support.CreateSupportPinned

    If (SelNodesNo > 0) Then
        ReDim SelNodes(SelNodesNo - 1) As Long

        'Get selected nodes
        objOpenSTAAD.Geometry.GetSelectedNodes SelNodes, 1

        'Assign pinned support to nodes
        For I = 0 To UBound(SelNodes)
            assignSupport = objOpenSTAAD.Support.AssignSupportToNode(SelNodes(I), pinnedSupport)
        Next

        'Get Support Count
        supportCount = objOpenSTAAD.Support.GetSupportCount
        MsgBox supportCount, vbOKOnly

        'Create a new view
        objOpenSTAAD.View.CreateNewViewForSelections
    Else
        MsgBox "No nodes are currently selected.", vbOKOnly
    End If

    Set objOpenSTAAD = Nothing
End Sub
```

◆ GetSupportInformation()

```
VARIANT OSSupportUI::GetSupportInformation ( const VARIANT FAR & vaSupportNode,
                                             VARIANT FAR &      varReleaseSpec,
                                             VARIANT FAR &      varSpringSpec )
```

Gets support information for the specified node, including release specification and spring specification.

Parameters

- [in] **vaSupportNode** The supported node number ID.
- [out] **varReleaseSpec** Degrees of freedom: Release (= 1) or Fixed (= 0) or Spring (=-1) for FX, FY, FZ, MX, MY and MZ.(Type:int)
- [out] **varSpringSpec** The variable spring constants: KFX, KFY, KFZ, KMX, KMY and KMZ.(Type:double)

Return values

- 0 No support.
- 1 Pinned support.
- 2 Fixed support.
- 3 Fixed support with releases.
- 4 Enforced support.
- 5 Enforced support with releases.
- 6 Inclined support.
- 7 Footing foundation.
- 8 Elastic mat foundation.
- 9 Plate mat foundation.
- 10 MultiLinear spring support.
- 11 Generated pinned support.
- 12 Generated fixed support.
- 13 Generated fixed support with releases.
- 1 General error.

Example: VBA

```
Sub SupportInformation()
    Dim objOpenSTAAD As Object
    Dim SelNodesNo As Long
    Dim SelNodes() As Long
    Dim fixedSupport As Long
    Dim pinnedSupport As Long
    Dim fixedButSupport As Long
    Dim assignSupport As Variant
    Dim supportCount As Variant
    Dim supportType As Variant
    Dim lSupportNodesArray() As Long
    Dim i As Integer
    Dim releaseData(5) As Double
    Dim springData(5) As Double
    Dim supportSpecType As Variant
    Dim msg As String

    'Launch OpenSTAAD Object
    Set objOpenSTAAD = GetObject(, "StaadPro.OpenSTAAD")

    'Get no. of selected nodes
    SelNodesNo = objOpenSTAAD.Geometry.GetNoOfSelectedNodes

    'Create fixed support
    fixedSupport = objOpenSTAAD.Support.CreateSupportFixed

    'Create pinned support
    pinnedSupport = objOpenSTAAD.Support.CreateSupportPinned

    'Create FixedBut support
```

```

Dim release(5) As Double
Dim spring(5) As Double
fixedButSupport = objOpenSTAAD.Support.CreateSupportFixedBut(release, spring)

If (SelNodesNo > 0) Then
    ReDim SelNodes(SelNodesNo - 1) As Long

    'Get selected nodes
    objOpenSTAAD.Geometry.GetSelectedNodes SelNodes, 1

    'Assign pinned support to nodes
    For i = 0 To UBound(SelNodes)
        assignSupport = objOpenSTAAD.Support.AssignSupportToNode(SelNodes(i), pinnedSupport)
        'Get Support Type : 1=Pinned; 2=Fixed; 3=FixedBut
        supportType = objOpenSTAAD.Support.GetSupportType(SelNodes(i))
        MsgBox supportType & " [1=Pinned; 2=Fixed; 3=FixedBut]", vbOKOnly

        'Gets Support Information
        supportSpecType = objOpenSTAAD.Support.GetSupportInformation(SelNodes(i), releaseData, springData)
        'Format(releaseData(1), "Standard")
        msg = Printf(" ReleaseData = [ {0}; {1}; {2}; {3}; {4}; {5}]", _
            Round(releaseData(0), 1), Round(releaseData(1), 1), Round(releaseData(2), 1), Round(releaseData(3),
1), Round(releaseData(4), 1), Round(releaseData(5), 1))
        MsgBox supportSpecType & msg, vbOKOnly
        msg = Printf(" SpringData = [ {0}; {1}; {2}; {3}; {4}; {5}]", _
            springData(0), springData(1), springData(2), springData(3), springData(4), springData(5))
        MsgBox supportSpecType & msg, vbOKOnly
    Next

    'Assign pinned support to nodes
    'assignSupport = objOpenSTAAD.Support.AssignSupportToNode(SelNodes, pinnedSupport)

    'Get Support Nodes
    ReDim lSupportNodesArray(0 To (objOpenSTAAD.Support.GetSupportCount - 1)) As Long
    Dim nodeArr As Variant
    nodeArr = objOpenSTAAD.Support.GetSupportNodes(lSupportNodesArray)
    MsgBox nodeArr, vbOKOnly

    'Get Support Count
    supportCount = objOpenSTAAD.Support.GetSupportCount
    MsgBox supportCount, vbOKOnly

    'Create a new view
    objOpenSTAAD.View.CreateNewViewForSelections
Else
    MsgBox "No nodes are currently selected.", vbOKOnly
End If

Set objOpenSTAAD = Nothing
End Sub

'Printf("Some text {0}, more text: {1}", A1, A2)
Public Function Printf(mask As String, ParamArray tokens()) As String
    Dim i As Long
    For i = 0 To UBound(tokens)
        mask = Replace$(mask, "{" & i & "}", tokens(i))
    Next
    Printf = mask
End Function

```

See also[OSSupportUI::GetSupportType](#)**◆ GetSupportInformationEx()**

```
VARIANT OSSupportUI::GetSupportInformationEx ( const VARIANT FAR & vaSupportNode,
                                              VARIANT FAR &      varSupportNo,
                                              VARIANT FAR &      varSupportType,
                                              VARIANT FAR &      varReleaseSpec,
                                              VARIANT FAR &      varSpringSpec )
```

Gets support information for the specified node, including release specification and spring specification.

Parameters

- [in] **vaSupportNode** The supported node number ID.
- [in] **varSupportNo** The support reference number
- [in] **varSupportType** The support type code
- [out] **varReleaseSpec** Degrees of freedom: Release (= 1) or Fixed (= 0) or Spring (=-1) for FX, FY, FZ, MX, MY and MZ.(Type:int)
- [out] **varSpringSpec** The variable spring constants: KFX, KFY, KFZ, KMX, KMY and KMZ.(Type:double)

Return values

- 0 No support.
- 1 Pinned support.
- 2 Fixed support.
- 3 Fixed support with releases.
- 4 Enforced support.
- 5 Enforced support with releases.
- 6 Inclined support.
- 7 Footing foundation.
- 8 Elastic mat foundation.
- 9 Plate mat foundation.
- 10 MultiLinear spring support.
- 11 Generated pinned support.
- 12 Generated fixed support.
- 13 Generated fixed support with releases.
- 1 General error.

Example: VBA

```
Sub SupportInformationExtended()
    Dim objOpenSTAAD As Object
    Dim SelNodesNo As Long
    Dim SelNodes() As Long
    Dim fixedSupport As Long
    Dim pinnedSupport As Long
    Dim fixedButSupport As Long
    Dim assignSupport As Variant
    Dim supportCount As Variant
    Dim supportType As Variant
    Dim lSupportNodesArray() As Long
    Dim i As Integer
    Dim releaseData(5) As Double
    Dim springData(5) As Double
    Dim supportSpecType As Variant
    Dim msg As String

    'Launch OpenSTAAD Object
    Set objOpenSTAAD = GetObject(, "StaadPro.OpenSTAAD")

    'Get no. of selected nodes
    SelNodesNo = objOpenSTAAD.Geometry.GetNoOfSelectedNodes    'SelNodesNo = Cells(2, 1)
```

```

'Create fixed support
fixedSupport = objOpenSTAAD.Support.CreateSupportFixed

'Create pinned support
pinnedSupport = objOpenSTAAD.Support.CreateSupportPinned

'Create FixedBut support
Dim release(5) As Double
Dim spring(5) As Double
fixedButSupport = objOpenSTAAD.Support.CreateSupportFixedBut(release, spring)

If (SelNodesNo > 0) Then
    ReDim SelNodes(SelNodesNo - 1) As Long

    'Get selected nodes
    objOpenSTAAD.Geometry.GetSelectedNodes SelNodes, 1

    'Assign pinned support to nodes
    For i = 0 To UBound(SelNodes)
        assignSupport = objOpenSTAAD.Support.AssignSupportToNode(SelNodes(i), pinnedSupport)
        'Get Support Type : 1=Pinned; 2=Fixed; 3=FixedBut
        supportType = objOpenSTAAD.Support.GetSupportType(SelNodes(i))
        MsgBox supportType & " [1=Pinned; 2=Fixed; 3=FixedBut]", vbOKOnly

        'Gets Extended Support Information
        supportSpecType = objOpenSTAAD.Support.GetSupportInformationEx(SelNodes(i), pinnedSupport,
        supportType, releaseData, springData)
        'Format(releaseData(1), "Standard")
        msg = Printf(" ReleaseData = [ {0}; {1}; {2}; {3}; {4}; {5}]", _
        Round(releaseData(0), 1), Round(releaseData(1), 1), Round(releaseData(2), 1), Round(releaseData(3),
        1), Round(releaseData(4), 1), Round(releaseData(5), 1))
        MsgBox supportSpecType & msg, vbOKOnly
        msg = Printf(" SpringData = [ {0}; {1}; {2}; {3}; {4}; {5}]", _
        springData(0), springData(1), springData(2), springData(3), springData(4), springData(5))
        MsgBox supportSpecType & msg, vbOKOnly
    Next

    'Assign pinned support to nodes
    assignSupport = objOpenSTAAD.Support.AssignSupportToNode(SelNodes, pinnedSupport)

    'Get Support Nodes
    ReDim lSupportNodesArray(0 To (objOpenSTAAD.Support.GetSupportCount - 1)) As Long
    Dim nodeArr As Variant
    nodeArr = objOpenSTAAD.Support.GetSupportNodes(lSupportNodesArray)
    MsgBox nodeArr, vbOKOnly

    'Get Support Count
    supportCount = objOpenSTAAD.Support.GetSupportCount
    MsgBox supportCount, vbOKOnly

    'Create a new view
    objOpenSTAAD.View.CreateNewViewForSelections
Else
    MsgBox "No nodes are currently selected.", vbOKOnly
End If

Set objOpenSTAAD = Nothing
End Sub

'Printf("Some text {0}, more text: {1}", A1, A2)
Public Function Printf(mask As String, ParamArray tokens()) As String
    Dim i As Long
    For i = 0 To UBound(tokens)
        mask = Replace$(mask, "{" & i & "}", tokens(i))
    Next
    Printf = mask
End Function

```

See also[OSSupportUI::GetSupportType](#)

◆ GetSupportNodes()

VARIANT OSSupportUI::GetSupportNodes (VARIANT FAR & varNodeNos)

Gets all supported nodes in an array.

Parameters

[out] **varNodeNos** The supported node number ID(s) VARIANT array.

Return values

<val> The number of supported node(s).

-1 General error.

Example: VBA

```
Sub SupportNodes()
    Dim objOpenSTAAD As Object
    Dim SelNodesNo As Long
    Dim SelNodes() As Long
    Dim pinnedSupport As Long
    Dim assignSupport As Variant
    Dim lSupportNodesArray() As Long
    Dim I As Integer

    'Launch OpenSTAAD Object
    Set objOpenSTAAD = GetObject(, "StaadPro.OpenSTAAD")

    'Get no. of selected nodes
    SelNodesNo = objOpenSTAAD.Geometry.GetNoOfSelectedNodes

    'Create pinned support
    pinnedSupport = objOpenSTAAD.Support.CreateSupportPinned

    If (SelNodesNo > 0) Then
        ReDim SelNodes(SelNodesNo - 1) As Long

        'Get selected nodes
        objOpenSTAAD.Geometry.GetSelectedNodes SelNodes, 1

        'Assign pinned support to nodes
        For I = 0 To UBound(SelNodes)
            assignSupport = objOpenSTAAD.Support.AssignSupportToNode(SelNodes(I), pinnedSupport)
        Next

        'Get Support Nodes
        ReDim lSupportNodesArray(0 To (objOpenSTAAD.Support.GetSupportCount - 1)) As Long
        Dim nodeArr As Variant
        nodeArr = objOpenSTAAD.Support.GetSupportNodes(lSupportNodesArray)
        MsgBox nodeArr, vbOKOnly

        'Create a new view
        objOpenSTAAD.View.CreateNewViewForSelections
    Else
        MsgBox "No nodes are currently selected.", vbOKOnly
    End If

    Set objOpenSTAAD = Nothing
End Sub
```

◆ GetSupportType()

VARIANT OSSupportUI::GetSupportType (VARIANT FAR & vaSupportNode)

Gets the support type for the specified node.

Parameters

[in] **vaSupportNode** The supported node number ID.

Return values

- 0 No support.
- 1 Pinned support.
- 2 Fixed support.
- 3 Fixed support with releases.
- 4 Enforced support.
- 5 Enforced support with releases.
- 6 Inclined support.
- 7 Footing foundation.
- 8 Elastic mat foundation.
- 9 Plate mat foundation.
- 10 MultiLinear spring support.
- 11 Generated pinned support.
- 12 Generated fixed support.
- 13 Generated fixed support with releases.
- 1 General error.

Example: VBA

```
Sub SupportType()
    Dim objOpenSTAAD As Object
    Dim SelNodesNo As Long
    Dim SelNodes() As Long
    Dim pinnedSupport As Long
    Dim assignSupport As Variant
    Dim supportType As Variant
    Dim I As Integer

    'Launch OpenSTAAD Object
    Set objOpenSTAAD = GetObject(, "StaadPro.OpenSTAAD")

    'Get no. of selected nodes
    SelNodesNo = objOpenSTAAD.Geometry.GetNoOfSelectedNodes

    'Create pinned support
    pinnedSupport = objOpenSTAAD.Support.CreateSupportPinned

    If (SelNodesNo > 0) Then
        ReDim SelNodes(SelNodesNo - 1) As Long

        'Get selected nodes
        objOpenSTAAD.Geometry.GetSelectedNodes SelNodes, 1

        'Assign pinned support to nodes
        For I = 0 To UBound(SelNodes)
            assignSupport = objOpenSTAAD.Support.AssignSupportToNode(SelNodes(I), pinnedSupport)

            'Get Support Type : 1=Pinned; 2=Fixed; 3=FixedBut
            supportType = objOpenSTAAD.Support.GetSupportType(SelNodes(I))
            MsgBox supportType, vbOKOnly
        Next

        'Create a new view
        objOpenSTAAD.View.CreateNewViewForSelections
    End If
End Sub
```

```
Else
  MsgBox "No nodes are currently selected.", vbOKOnly
End If

Set objOpenSTAAD = Nothing
End Sub
```

See also[OSSupportUI::GetSupportInformation](#)

© Copyright Bentley Systems, Inc. For more information, see <http://www.bentley.com>.